

# Chemistry–A European Journal

Supporting Information

## **Galectin–Glycan Interactions: Guidelines for Monitoring by $^{77}\text{Se}$ NMR Spectroscopy, and Solvent ( $\text{H}_2\text{O}/\text{D}_2\text{O}$ ) Impact on Binding**

Tammo Diercks<sup>+,\*[a]</sup> Francisco J. Medrano<sup>+, [b]</sup> Forrest G. FitzGerald<sup>+, [c]</sup> Donella Beckwith<sup>+, [c]</sup>  
Martin Jaeger Pedersen,<sup>[d]</sup> Mark Reihill,<sup>[d]</sup> Anna-Kristin Ludwig<sup>+, [e]</sup> Antonio Romero,<sup>\*,[b]</sup>  
Stefan Oscarson,<sup>\*,[d]</sup> Maré Cudic,<sup>\*,[c]</sup> and Hans-Joachim Gabius<sup>\*,[e]</sup>

## Contents

NMR pulse programs .....	2
2D $^1\text{H},^{77}\text{Se}$ HSQMBC-CPMG .....	2
2D $^1\text{H},^{77}\text{Se}$ HeHaHa (heteronuclear Hartmann-Hahn transfer) .....	5
1D $^1\text{H}$ spectra of SeDG with/out $^{77}\text{Se}$ decoupling .....	9
$^1\text{H}\rightarrow^{77}\text{Se}$ correlation: transfer intensity versus transfer delay (SeDG) .....	9
2D $^1\text{H},^{77}\text{Se}$ NMR correlation spectra .....	10
1D $^1\text{H}[^{77}\text{Se}]$ traces for SeDGlc .....	11
Fluorescence anisotropy .....	12
CD spectroscopy.....	13
ITC Titration Profiles .....	15
Synthesis of DiLacNAc and LNT tetrasaccharides .....	18

## NMR pulse programs

The following NMR pulse programs for 2D  $^1\text{H}$ ,  $^{77}\text{Se}$  long-range correlation were written and tested under TopSpin 3.2.7 on a BRUKER AVANCE III 600 MHz spectrometer. Auxiliary files (e.g., for shaped pulses and CPD sequences) can be obtained from [tdiercks@cicbioqune.es](mailto:tdiercks@cicbioqune.es) upon request.

### 2D $^1\text{H}$ , $^{77}\text{Se}$ HSQMBC-CPMG

```
;tdHSQMBCcpmg.gs ;17/05/2019

##### adapted for TopSpin 3 and AVANCE III #####

;2D H,X HSQMBC (long-range correlation) using CPMG-INEPT
;- CPMG-INEPT to enforce inphase (TOCSY) evolution of competing nJHH coupling
; using XY16 phase cycling (i.e. MOCCA mixing on H)
; NOTE: ### XY16-CPMG bandwidth BW_XY16 depends on 2 criteria: ###
; ### a) BW_XY16 <= Brf/4 = 1/(16*p10) ###
; ### b) BW_XY16 <= Bcpmg = cnst15 ###
;- with coherence selection by echo/antiecho gradients
;- further on-resonant water suppression by presaturation (plw9) and flip-back
;- with pre-scan X depolarisation to reduce t1 noise
;-Dd4opt: optimise evolution delay d4 in steps of 5ms
; Analysis: optimal d4 = 5ms * row_max

;----- Processing info -----
;Dimension: FnMode WDW SSB REVERSE FCOR PHC0 PHC1
;F1(X) E/A QSINE 2 FALSE 0.5 0 0

;----- Parameter settings -----
;## DIGMOD = baseopt ###
;ns = ## 8*n (4*n: minimal F1 artefacts) ##
;zgoptns = -Dd4opt
;cnst15 = computed Bcpmg [Hz] = BANDWIDTH
;l1 = X decoupling in F2: 0(OFF),1(ON)
;--- 1H (f1) ---
;sfo1 = ### H2O on-resonant ###
;p1 = 90deg on 1H (@ pl1)
;pl1 = for p1=90deg on 1H
;p2 <= 200us*600/BF1 (power check on sp2!)
;spnam2 = ### BIP_720_(50 or 75).. ###
;pl9 < 20uW (optional presaturation, 0 = OFF)
;--- X (f2) ---
;p3 = 90deg on X (@pl2)
;pl2 = for p3=90deg on X
;cnst62 = bandwidth[ppm] of X dec. in F2 (UNFOLDED WINDOW)
;cpdprg2 = ### GARP4.p62 ###
;---- Delays ---
;d1 = relaxation delay (~ 1.25 T1(H))
;d4 < 0.5/nJHX
;l4 = sets desired echo delay d40 & bandwidth = cnst15
;d16 > 100 us (eddy current recovery)
;d1 = relaxation delay (~ 1.25 T1(H))
;----- GRADIENTS -----
;p16 >= 400us (gradient pulse)
;gpnam1 = SINE.50 or SMSQ10.50
;gpz1 > 70 (strong z-spoil)
```

```
;gpnam4 = SINE.100 or SMSQ10.50
;gpz4 > 50 (strong z-spoil)
```

```
;----- calculated parameters -----
```

```
#include <Grad.incl>
#include <Avance.incl>
;--- 1H pulses ---
#define H f1
"plw10=plw1*pow(p1/p10,2)"
"spoff2=0" ;for 1H BIP on ALL 1H
"spaal2=0.5"
"spw2 = plw1*pow(8*p1/p2,2)" ;for p2 = BIP_720_...
;--- X pulses ---
#define X f2
"p62=1000000/(bf2*cnst62)" ;for GARP4 decoupling
if(l1)
{;--- with X decoupling in F2 ---
"plw12 = plw2*pow(p3/p62,2)" ;for p62 = 90deg square pulse
;--- 15N pulses ---
}
else
{
"plw12=0"
}
"plw20=plw2*pow(p3/p10,2)"
;--- delays ---
define delay d0c
define delay d1c
define delay d12c
define delay d40c
define delay comp
define delay prompt
"cnst16=sfo2/sfo1"
"d11=10m"
"d12=5u"
"in0=inf1/2"
"d0=4u"
"d0c=d0*2+p2"
#ifdef d4opt
"l4=1"
"d4=5m"
#endif
"l14=(l4*16)-2"
"d40=d4/(4+2*l14)-p10"
"d40c=d40-p10*2/PI"
"comp=p10*4/PI+d12+2*(p16+d16+p3)+d0c"
"cnst15=250000/(p10+d40)" ;Bcpmg field strength (Hz)
"d1c=d1-(aq+d11+3m+d12*2)"
"prompt=1u+cnst15*1u+l1*1u"
"acqt0 = -d12"

1 ze
prompt
2 d11 do:X
3 d12 pl9:H pl2:X
d1c cw:H
d12 do:H
```

```

(p3 ph0):X ;depolarise X prior to scan to reduce t1 noise
d12 UNBLKGRAD
2mp:gp4 ;z-spoil
0.7m pl10:H pl20:X
0.3m rpp20
;---- CPMG-INEPT using XY16 -----
(p10 ph1):H
d40c
(p10*2 ph20):H (p10*2 ph20^):X
d40
14 d40
(p10*2 ph20):H (p10*2 ph20^):X
d40
lo to 14 times l14
d40
(p10*2 ph20):H (p10*2 ph20^):X
d40c
(p10 ph12):H
;---- z-spoil and H2Ox dephasing ----
d12
(p2:sp2 ph0):H ;compensating BIP
1.3mp:gp4
0.5m pl1:H pl2:X
0.2m rpp20
;---- t1(X) ----
(p3 ph5):X
d0
(p2:sp2 ph1):H ;invert H2O to -x
d0
p16:gp1*EA
d16
(p3*2 ph4):X
p16:gp1*EA*-1
d16
d0c
(p3 ph15):X
;---- z-spoil and H2O-x rephasing ----
1.3mp:gp4*-1
0.7m pl10:H pl20:X
comp
;---- CPMG-INEPT using XY16 -----
(p10 ph1):H ;flips H2O-x to +z
d40c
(p10*2 ph20):H (p10*2 ph20^):X
d40
24 d40
(p10*2 ph20):H (p10*2 ph20^):X
d40
lo to 24 times l14
d40
(p10*2 ph20):H (p10*2 ph20^):X
d40
;--- decoding gradient echo ---
p16:gp1*cnst16*-0.93
d16 pl1:H
(p1*2 ph0):H
p16:gp1*cnst16*1.07

```

```

d16 pl12:X
d12 BLKGRAD
go=2 ph31 cpd2:X
10u do:X
20u cpds2:X ;reset decoupler pointer after complete complex point
10u do:X
d11 mc #0 to 2

#ifdef d4opt
F1QF(calclc(l14,+16))
#endif

#ifndef d4opt
F1EA(calgrad(EA), caldel(d0,+in0) & calph(ph5,+180) & calph(ph31,+180))
#endif
exit

ph0= 0
ph1= 1
ph2= 2
ph3= 3
ph4= 1
ph5= 0 2 ;(X) excitation pulse with axial peak suppression in t1
ph12={{2}*4}^2
ph15= 0 0 2 2 ;(X) read-out pulse after t1 (CRITICAL)
ph20= 0 1 0 1 1 0 1 0 2 3 2 3 3 2 3 2 ;XY16 cycle
ph31={0 2 2 0}^2

```

## 2D <sup>1</sup>H, <sup>77</sup>Se HeHaHa (heteronuclear Hartmann-Hahn transfer)

Depending on the DIPS12 field strength, this experiment achieves broadband or narrow-band (selective) HeHaHa transfer. For this, parameter “cnst15” must be set as described below.

```

;tdHXhehaha.gs ;16/05/2019
;#### adapted for TopSpin 3 and AVANCE III ###

```

```

;2D H,X hetero-TOCSY (Hetero Hartmann-Hahn transfer)
;- with DIPS12 heHaHa to simultaneously enforce pure inphase evolution of
; competing nJHH coupling (for pure phase long-range H,X transfer)
;- with coherence selection by echo/antiecho gradients
;- with universal flip-back
;- with optional on-resonant presaturation (during d1, power plw9)
;- with pre-scan X depolarisation to reduce t1 noise
;- zgoption: -Dl15opt to optimise l15 in steps of 'dDIPS12'
;##### Important Set-up NOTES #####
;- Set-up for DIPS12 mixing:
;1. Define bandwidth for HeHaHa:
; cnst15[Hz] = 2*(most distant signal H or X from offset H or X)
;2. Define mixing time d15
; Set l15 such that resulting d15 is ca. 0.5/J(CX)
; Possibly increase cnst15 (-> decreases p10) to fine-adjust d15!
;- Set-up for l15 (mixing time) optimisation:
; set zgoptn = -Dl15opt to measure l15 dependent transfer efficiency
; set FnMode = QF, TD1 = S11 = 8-16, possibly increase cnst15
; process by 'xf2' and select row_max with maximal transfer
; -> Corresponding mixing time: d15 = dDIPS12*row_max

```

```

;----- Processing info -----
;Dimension: FnMode WDW SSB REVERSE FCOR PHCO PHC1
;F1(X) E/A QSINE 2 FALSE 0.5 0 0

;----- Parameter settings -----
;## DIGMOD = baseopt ###
;ns = ## 8*n (4*n: minimal F1 artefacts) ##
;zgoptns = -Dl15opt
;l1 = X decoupling in F2:0(OFF),1(ON)
;--- 1H (f1) ---
;sfo1 = ## center of H for HeHaHa mixing ##
;cnst1 = H2O offset for presaturation (o1 from zgpr)
;p1 = 90deg on 1H (@ pl1)
;pl1 = for p1=90deg on 1H
;p2 <= 200us*600/BF1 (power check on sp2!)
;spnam2 = ### BIP_720_(50 or 75).. ###
;p10 ~ 1/(4*BW) [BW = desired bandwidth in Hz]
;--- X (f2) ---
;sfo2 = ## center of X for HeHaHa mixing ##
;p3 = 90deg on X (@pl2)
;pl2 = for p3=90deg on X
;cnst62 = bandwidth[ppm] of X dec. in F2 (UNFOLDED WINDOW)
;cpdprg2 = ### GARP4.p62 ###
;p62 = [600/bf1]*75u(Call),*120u(Caliph),*85u(Carom)
;---- Delays ---
;d1 = relaxation delay (~ 1.25 T1(H))
;cnst15 = minimal BW[Hz] for mixing
;l15 = DIPSI2 loops for desired d15 = mixing time
;d15 = computed mixing time (= < 0.5/nJHX)
;d16 > 100 us (eddy current recovery)
;----- GRADIENTS -----
;p16 >= 400us (gradient pulse length)
;gpnam1 = SINE.50
;gpz1 = 73 (coherence selection gradient)
;gpnam4 = SINE.100
;gpz4 = 47.31 (z-spoil gradient)

;----- calculated parameters -----
#include <Grad.incl>
#include <Avance.incl>
;--- 1H pulses ---
#define H f1
;"cnst10=cnst1-o1"
"p10=1s/(4*cnst15)"
"plw10=plw1*pow(p1/p10,2)"
"spoff2=0" ;for 1H BIP on ALL 1H
"spaal2=0.5"
"spw2 = plw1*pow(8*p1/p2,2)" ;for p2 = BIP_720_...
;--- X pulses ---
#define X f2
"p62=1000000/(bf2*cnst62)" ;for GARP4 decoupling
if(l1)
{;--- with X decoupling ---
"plw12 = plw2*pow(p3/p62,2)" ;for p62 = 90deg square pulse
;--- 15N pulses ---
}
else

```

```

{;--- without X decoupling ---
"plw12 = 0"
}
"plw20=plw2*pow(p3/p10,2)"
;--- delays ---
define delay d0c
define delay d1c
define delay dDIPSI2
define delay prompt
"cnst16=sfo2/sfo1"
"dDIPSI2=28.78*4*p10"
"d11=10m"
"d12=5u"
"in0=inf1/2"
"d0=4u"
"d0c=d0*2+p2"
"d1c=d1-(aq+d11)"
#ifdef l15opt
"l15=1"
#endif
"d15=l15*dDIPSI2"
#define DIPSI2a p10*3.556 ph20 p10*4.556 ph22 p10*3.222 ph20 p10*3.167 ph22 p10*0.333 ph20 p10*2.722
ph22 p10*4.167 ph20 p10*2.944 ph22 p10*4.111 ph20
#define DIPSI2b p10*3.556 ph22 p10*4.556 ph20 p10*3.222 ph22 p10*3.167 ph20 p10*0.333 ph22 p10*2.722
ph20 p10*4.167 ph22 p10*2.944 ph20 p10*4.111 ph22

"prompt = d12+l1*1u"
"acqt0 = 0"

1 ze
  prompt
2 d11 do:X
3 d12 pl9:H pl2:X
  1m fq=cnst1 (bf):H           ;jump to H2O for optional presaturation
  d1c cw:H
  d12 do:H
  1m fq=0:H                   ;jump to H center of HeHaHa mixing
  (p3 ph0):X                 ;depolarise X prior to scan to reduce t1 noise
  d12 UNBLKGRAD
  2mp:gp4                     ;z-spoil
  1m pl10:H pl20:X
;---- HeHaHa -----
  (p10 ph1):H                 ;create Hx for mixing
15 (DIPSI2a):H (DIPSI2a):X    ;DIPSI2 along x
  (DIPSI2b):H (DIPSI2b):X    ;DIPSI2 along -x
  (DIPSI2b):H (DIPSI2b):X    ;DIPSI2 along -x
  (DIPSI2a):H (DIPSI2a):X    ;DIPSI2 along x
  lo to 15 times l15
  (p10 ph1):H (p10 ph3):X    ;flip back Hx->-Hz and Xx->Xz
;---- z-spoil, unused Hu in -z ----
  1.7mp:gp4                   ;suppress radiation damping on H2O-z
  1.3m pl1:H pl2:X
;---- t1(X) ----
  (p3 ph5):X
  d0
  (p2:sp2 ph1):H             ;flip Hu to +z
  d0

```




```

p16:gp1*EA
d16
(p3*2 ph4):X
p16:gp1*EA*-1
d16
d0c
(p3 ph15):X
;---- z-spoil ----
(p2:sp2 ph1):H ;flip Hu forward to -z
1.3mp:gp4
1m pl10:H pl20:X
;---- HeHaHa to Hx -----
(p10 ph25):X ;create Xx for mixing
16 (DIPSI2a):H (DIPSI2a):X ;DIPSI2 along x
(DIPSI2b):H (DIPSI2b):X ;DIPSI2 along -x
(DIPSI2b):H (DIPSI2b):X ;DIPSI2 along -x
(DIPSI2a):H (DIPSI2a):X ;DIPSI2 along x
lo to 16 times l15
p16:gp1*cnst16*-0.93
d16 pl1:H pl12:X
d12
(p1*2 ph0):H ;universal flip-back
p16:gp1*cnst16*1.07
d16
d12 BLKGRAD
go=2 ph31 cpd2:X
10u do:X
20u cpds2:X ;reset decoupler pointer after complete complex points
10u do:X
d11 mc #0 to 2
#ifdef l15opt
F1QF(calclc(l15,+1))
#endif
#ifndef l15opt
F1EA(calgrad(EA), caldel(d0,+in0) & calph(ph5,+180) & calph(ph31,+180))
#endif
exit

ph0= 0
ph1= 1
ph2= 2
ph3= 3
ph4= {{1}*4}^1 ;(X) rephasing pulse in t1
ph5= 0 2 ;(X) excitation pulse with axial peak suppression in t1
ph15= {{0}*8}^2 ;(X) read-out pulse after t1 (CRITICAL)
ph25= {{1}*2}^2 ;(X) reTOCSY excitation pulse
ph20= 0 ;dipsi phase 1
ph22= 2 ;dipsi phase 2
ph31={{0 2 2 0}^2}^2


```

## 1D $^1\text{H}$ spectra of SeDG with/out $^{77}\text{Se}$ decoupling



**Figure S1:** 1D  $^1\text{H}$  spectrum of SeDG with (red) and without (blue)  $^{77}\text{Se}$  decoupling, evincing the indicated direct n-bond  $^n\text{J}_{\text{H},\text{Se}}$  coupling constants.


## $^1\text{H} \rightarrow ^{77}\text{Se}$ correlation: transfer intensity versus transfer delay (SeDG)




**Figure S2.** Transfer time dependence of SeDGal H1 to H4 signal intensities for different heteronuclear  $^1\text{H}, ^{77}\text{Se}$  correlation schemes: (i) CPMG-INEPT (straight lines) with  $B_{\text{CPMG}} = 1371$  Hz,  $\delta_{\text{CPMG}} = 157.3$   $\mu\text{s}$ , (ii) broadband heteronuclear Hartmann-Hahn (HeHaHa) transfer by high-power DIPS12 mixing with  $B_{\text{DIPS12}} = 1450$  Hz (dotted lines), and (iii) H1 selective heteronuclear Hartmann-Hahn (sel. HeHaHa) transfer by low-power DIPS12 mixing with  $B_{\text{DIPS12}} = 285$  Hz centered on H1 (dashed-dotted lines).

## 2D $^1\text{H}$ , $^{77}\text{Se}$ NMR correlation spectra


All experiments were acquired at 298 K on a BRUKER AVIII 600 MHz spectrometer equipped with z-TBI probehead using the pulse programs printed above and identical total interscan recovery time ( $d1 = 1.5$  s), scan number ( $ns = 8$ ), FID resolution (1.2 Hz for  $^1\text{H}$ , 2 Hz for  $^{77}\text{Se}$ ), resulting in 0.5 h total measurement time. All spectra are processed and plotted identically.



**Figure S3a.** 2D  $^1\text{H}$ ,  $^{77}\text{Se}$  HSQMBC-CPMG with  $B_{\text{CPMG}}(\text{cnst15}) = 1143$  Hz at  $B_{\text{RF}} = 10.000$  Hz ( $p_{10} = 25$   $\mu\text{s}$ ),  $\Delta(d4) = 70$  ms.




**Figure S3b.** 2D  $^1\text{H}$ ,  $^{77}\text{Se}$  HeHaHa with  $B_{\text{DIPSI2}}(\text{cnst15}) = 1450$  Hz ( $p_{10} = 172.4$   $\mu\text{s}$ ),  $\Delta(d8) = 80$  ms.




**Figure S3c.** 2D  $^1\text{H},^{77}\text{Se}$  H1-selective HeHaHa with  $B_{\text{DIPSI2}}(\text{cnst15}) = 285$  Hz ( $p_{10} = 877.2$   $\mu\text{s}$ ) centered at 4.98 ppm (center of H1 for SeDG and SeDGlc),  $\Delta(d8) = 100\text{ms}$ .

### 1D $^1\text{H}[^{77}\text{Se}]$ traces for SeDGlc




**Figure S4.** 1D  $^1\text{H}$  traces from 2D  $^1\text{H},^{77}\text{Se}$  correlation spectra taken at the  $^{77}\text{Se}$  chemical shift of SeDGlc (394.25 ppm) in the absence (grey) or presence (orange) of hGal-3 (0.125 mM). **(A)** CPMG-HSQMBC ( $\Delta_{\text{opt}}(\text{H1}) = 70$  ms,  $B_{\text{CPMG}} = 1143$  Hz). The indirect H3 correlation signal ( $^4J_{\text{H3,Se}} = 0$ ) derives from HoHaHa (TOCSY) transfer. **(B)** HeHaHa ( $B_{\text{DIPSI2}} = 1450$  Hz,  $\Delta_{\text{opt}}(\text{H1}) = 80$  ms). **(C)** H1 selective HeHaHa ( $B_{\text{DIPSI2}} = 285$  Hz,  $\Delta_{\text{opt}}(\text{H1}) = 100$  ms). The small intensity reduction (ca. -10%) in the presence of hGal-3 is likely due to a dilution effect.

## Fluorescence anisotropy




**Figure S5.** Determination of hGal-3 CRD and hGal3-hGal3 interactions with increasing concentrations of TDG and SeDG by intrinsic Trp fluorescence anisotropy. Binding isotherms of TDG (A) and SeDG (B) to Gal-3 CRD, and of TDG (C) and SeDG (D) to Gal3-Gal3 at 25 °C. Equilibrium  $K_D$  values were determined as described in methods. Error bars represent standard deviations from five repeated anisotropy measurements. Errors of  $K_D$  represent error estimates from the fitting algorithm.

## CD spectroscopy




**Figure S6.** Interaction of Gal-3 CRD and Gal3-Gal3 with the ligands TDG and SeDG, observed by circular dichroism. **(A)** Near-UV CD spectra of Gal-3 CRD in the absence (black) and in the presence of increasing amounts of TDG (red, 31.0  $\mu\text{M}$ ; green, 61.8  $\mu\text{M}$ ; blue, 92.5  $\mu\text{M}$ ; cyan, 122.9  $\mu\text{M}$  and magenta, 301.7  $\mu\text{M}$ ). Inset: binding isotherms of TDG to Gal-3 CRD at 25 °C. **(B)** Near-UV CD spectra of Gal-3 CRD in the absence (black) and in the presence of increasing amounts of SeDG (red, 37.6  $\mu\text{M}$ ; green, 75.0  $\mu\text{M}$ ; blue, 112.1  $\mu\text{M}$ ; cyan, 148.9  $\mu\text{M}$ ; magenta, 364.0  $\mu\text{M}$  and yellow, 570.0  $\mu\text{M}$ ). Inset: binding isotherm of SeDG to Gal-3 CRD at 25 °C. **(C)** Near-UV CD spectra of Gal3-Gal3 in the absence (black) and in the presence of increasing amounts of TDG (red, 24.9  $\mu\text{M}$ ; green, 49.8  $\mu\text{M}$ ; blue, 99.0  $\mu\text{M}$ ; cyan, 147.8  $\mu\text{M}$ ; magenta, 243.9  $\mu\text{M}$  and yellow, 430.6  $\mu\text{M}$ ). Inset: binding isotherm of TDG to Gal3-Gal3 at 25 °C. **(D)** Near-UV CD spectra of Gal3-Gal3 in the absence (black) and in the presence of increasing amounts of SeDG (red, 27.7  $\mu\text{M}$ ; green, 55.2  $\mu\text{M}$ ; blue, 82.6  $\mu\text{M}$ ; cyan, 109.9  $\mu\text{M}$ ; magenta, 270.3  $\mu\text{M}$  and yellow, 425.5  $\mu\text{M}$ ). Inset: binding isotherm of SeDG to Gal3-Gal3 at 25 °C. Equilibrium  $K_D$  values were determined as described in methods. Error bars represent an estimation of the noise of the CD spectra. Errors of  $K_D$  represent error estimates from the fitting algorithm.



**Figure S7.** Effect of H<sub>2</sub>O vs. D<sub>2</sub>O on the interactions of Gal-1 with the ligands TDG and SeDG, observed by circular dichroism. **(A)** Near-UV CD spectra of Gal-1 in H<sub>2</sub>O in the absence (black) and in the presence of increasing amounts of TDG (red, 39.8  $\mu\text{M}$ ; green, 79.4  $\mu\text{M}$ ; blue, 118.6  $\mu\text{M}$ ; cyan, 157.5  $\mu\text{M}$  and magenta, 385.5  $\mu\text{M}$ ). Inset: binding isotherms of TDG to Gal-1 at 25 °C. **(B)** Near-UV CD spectra of Gal-1 in D<sub>2</sub>O in the absence (black) and in the presence of increasing amounts of TDG (red, 33.2  $\mu\text{M}$ ; green, 66.2  $\mu\text{M}$ ; blue, 99.0  $\mu\text{M}$ ; cyan, 131.6  $\mu\text{M}$ ; magenta, 322.6  $\mu\text{M}$  and yellow, 506.3  $\mu\text{M}$ ). Inset: binding isotherms of SeDG to Gal-1 at 25 °C. **(C)** Near-UV CD spectra of Gal-1 in H<sub>2</sub>O in the absence (black) and in the presence of increasing amounts of SeDG (red, 49.8  $\mu\text{M}$ ; green, 99.0  $\mu\text{M}$ ; blue, 147.8  $\mu\text{M}$ ; cyan, 196.1  $\mu\text{M}$  and magenta, 476.2  $\mu\text{M}$ ). Inset: binding isotherms of SeDG to Gal-1 at 25 °C. **(D)** Near-UV CD spectra of Gal-1 in D<sub>2</sub>O in the absence (black) and in the presence of increasing amounts of SeDG (red, 33.2  $\mu\text{M}$ ; green, 66.2  $\mu\text{M}$ ; blue, 99.0  $\mu\text{M}$ ; cyan, 131.6  $\mu\text{M}$ ; magenta, 322.6  $\mu\text{M}$  and yellow, 506.3  $\mu\text{M}$ ). Inset: binding isotherms of SeDG to Gal-1 at 25 °C. Equilibrium  $K_D$  values were determined as described in methods. Error bars represent an estimation of the noise of the CD spectra. Errors of  $K_D$  represent error estimates from the fitting algorithm.






## ITC Titration Profiles

ITC measurements were done on samples dissolved in phosphate buffer (pH 7.2) containing 20 mM phosphate, 10 mM NaCl, 10 mM BME, in either H<sub>2</sub>O or D<sub>2</sub>O. Ligands were injected every 150 s at 298 K. Data fitting to a one-site binding model used MicroCal PEAQ-ITC analysis software; the resulting values for stoichiometry ( $n$ ), binding affinity ( $K_a$ ), dissociation constant ( $K_d$ ), enthalpy ( $\Delta H$ ), and change in entropy with respect to temperature ( $T\Delta S$ ) are summarized in Table 3 of the article.




**Figure S8a.** ITC titration profiles for Gal-1 in H<sub>2</sub>O. (A) Gal-1 (110  $\mu\text{M}$ ) with TDG (3.2 mM), (B) Gal-1 (140  $\mu\text{M}$ ) with SeDG (8.0 mM), (C) Gal-1 (140  $\mu\text{M}$ ) with LacNAc (6.0 mM), (D) Gal-1 (171  $\mu\text{M}$ ) with DiLacNAc (6.0 mM), (E) Gal-1 (162  $\mu\text{M}$ ) with LNT (6.0 mM). Top panels: experimental ITC data. Bottom panels: data fitting to a one-site binding model.




**A) + TDG****B) + SeDG****C) + LacNAc****D) + DiLacNAc****E) + LNT**

**Figure S8b.** ITC titration profiles for Gal-1 in D<sub>2</sub>O. (A) Gal-1 (90 µM) with TDG (3.8 mM), (B) Gal-1 (120 µM) with SeDG (8.0 mM), (C) Gal-1 (110 µM) with LacNAc (6.0 mM), (D) Gal-1 (120 µM) with DiLacNAc (6.0 mM), (E) Gal-1 (152 µM) with LNT (6.0 mM). Top panels: experimental ITC data. Bottom panels: data fitting to a one-site binding model.


A) + TDG



B) + LacNAc




C) + LNT




**Figure S9a.** ITC titration profiles for Gal-3 in H<sub>2</sub>O. (A) Gal-3 (130 μM) with TDG (3.8 mM), (C) Gal-3 (130 μM) with LacNAc (6.0 mM), (D) Gal-3 (150 μM) with LNT (6.0 mM). Top panels: experimental ITC data. Bottom panels: data fitting to a one-site binding model.


A) + TDG




B) + SeDG



C) + LacNAc




D) + LNT



**Figure S9b.** ITC titration profiles for Gal-3 in D<sub>2</sub>O. (A) Gal-3 (45 μM) with TDG (3.8 mM), (B) Gal-3 (45 μM) with SeDG (6.0 mM), (C) Gal-3 (45 μM) with LacNAc (6.0 mM), (D) Gal-3 (75 μM) with LNT (1.0 mM). Top panels: experimental ITC data. Bottom panels: data fitting to a one-site binding model

## Synthesis of DiLacNAc and LNT tetrasaccharides

The DiLacNAc tetrasaccharide **1** and the LNT tetrasaccharide **2** were synthesized through deprotection of tetrasaccharides **3**<sup>1</sup> and **5**<sup>3</sup>, respectively (Scheme). The benzyl groups in DiLacNAc precursor **3**, efficiently obtained using a trifluoroacetamide protected lactosamine thioglycoside donor and a benzylated LacNAc acceptor, were removed under oxidative conditions which do not reduce the azide.<sup>2</sup> Following base treatment and peracetylation afforded compound **4**, which was deacetylated under Zemplen conditions to give target DiLacNAc compound **1**. The benzyl groups in the LNT precursor **5**, synthesized as an intermediate for continued Lewis x and H antigen synthesis,<sup>3</sup> were removed using catalytic hydrogenolysis, which also reduced the azide to an amino group, to afford target LNT compound **2**.



**Scheme S1** Synthesis of DiLacNAc compound **1** and LNT compound **2**. Reagents: (a) i. NaBrO<sub>3</sub>, Na<sub>2</sub>S<sub>2</sub>O<sub>4</sub>, H<sub>2</sub>O, AcOEt, ii. NaOMe, MeOH, iii. Ac<sub>2</sub>O, pyridine; (b) NaOMe, DMSO, MeOH; (c) H<sub>2</sub>, Pd/C, Pd(OH)<sub>2</sub>/C, HCl (aq), H<sub>2</sub>O, EtOH, EtOAc.

### General Methods

Reactions were monitored by thin-layer chromatography (TLC) on Merck DC-Alufolien plates precoated with silica gel 60 F<sub>254</sub>. Visualisation was performed with UV-light (254 nm) and/or staining with 8% H<sub>2</sub>SO<sub>4</sub>/EtOH solution. All chemicals were purchased from commercial suppliers (Acros, Carbosynth Ltd., Fisher Scientific Ltd., Glycom A/S, Merck, Sigma-Aldrich, VWR) and were used without purification. Dry solvents were obtained from a PureSolv-EN<sup>TM</sup> solvent purification system (Innovative Technology Inc.) or were used as purchased from Sigma-Aldrich in AcroSeal<sup>®</sup> bottles. NMR spectra were recorded on Varian Inova spectrometers at 25 °C. High-resolution mass spectrometry (HRMS) data were recorded on a Waters Micromass LCT LC-TOF instrument using electrospray ionisation (ESI) in positive mode. Specific rotations were recorded on a Perkin-Elmer polarimeter (Model 343) at the sodium D-line (589 nm) at 20 °C in a 1 dm cell. Deprotected sugars were lyophilised using an Alpha 1-2 LDplus (Christ Ltd.) freeze-dryer: pressure: 0.035 mbar; ice-condenser temperature: -55 °C.

### References:

1. C. Romanò, S. Oscarson, *Org. Biomol. Chem.*, **2019**, 17, 2265-2278.
2. M. Niemietz, L. Perkams, J. Hoffman, S. Eller, C. Unverzagt, Y. Ito, M. Niemietz, M. Pischl and C. Raps, *Chem. Commun.*, **2011**, 47, 10485.
3. M. Lahmann, L. Bülow, P. Teodorovic, H. Gybäck, and S. Oscarson, *Glycoconj. J.*, **2004**, 21, 251-256.




Figure S10A. <sup>1</sup>H-NMR of compound 1 (298 K, in D<sub>2</sub>O, 500 MHz)




Figure S10B. <sup>13</sup>C-NMR of compound 1 (298 K, in D<sub>2</sub>O, 500 MHz)




Figure S10C. CH<sub>n</sub> multiplicity edited 2D <sup>1</sup>H,<sup>13</sup>C HSQC of compound 1.






Figure S10D. 2D <sup>1</sup>H,<sup>1</sup>H COSY spectrum (magnitude mode) of compound 1



**Figure S11A.** 1D  $^1\text{H}$  NMR spectrum of compound 2



**Figure S11B.** CH<sub>n</sub> multiplicity edited 2D  $^1\text{H}$ ,  $^{13}\text{C}$  HSQC spectrum of compound 2